
バイトコード変換によるActionScript

プログラムのマルチスレッド化

杉 本 卓 哉†1 新 城 靖†1 中 井 央†2

板 野 肯 三†1 佐 藤 聡†1

ActionScriptによるプログラミングでは C言語や Java言語などでのマルチスレッ
ドプログラミングに慣れた開発者にとって制御フローの記述について制約をかかえて
いる．ユーザ定義関数は以下の 3つのケースにおいて，処理の途中で処理系にリター
ンする必要がある．3つのケースとは，1つはサーバからの応答を受け取る場合，1つ
は画面の再描画の処理を行う場合，1 つは CPU 時間のかかる処理の際にWeb ブラ
ウザ全体のフリーズを回避する場合である．これらのケースにおいて処理の途中で処
理系にリターンするために，ActionScriptではイベント駆動プログラミングが行われ
る．しかし，イベント駆動によるプログラミングでは単純な処理も煩雑なプログラム
となってしまう．本研究では，以上の問題を解決するために ActionScript でマルチ
スレッドプログラミングを可能にする．提案手法では，まず開発者は本手法において
提供する API を用いてマルチスレッドのプログラムを記述し，それを標準のコンパ
イラでバイトコードにコンパイルする．次に，本手法において提供するコード変換器
を用いて，バイトコード中の各関数に実行状態を保存して処理を中断するコードを挿
入する．実行時には，本手法において提供するスレッド管理機構が最初の実行可能な
スレッドの関数を呼び出す．呼び出された関数はそれぞれの処理を実行するが，挿入
された中断コードが実行されるとスレッド管理機構にリターンする．スレッド管理機
構は次の実行可能なスレッドの関数を呼び出す．提案手法を利用し，アニメーション
やサーバからのファイルのダウンロードを行うプログラムをマルチスレッドで開発す
ることができた．

Multithreading ActionScript Programs
with Bytecode Translation

Takuya Sugimoto,†1 Yasushi Shinjo,†1

Hisashi Nakai,†2 Kozo Itano†1 and Akira Sato†1

Programming in ActionScript has restrictions on coding control flows for de-

velopers who are familiar with mutithread programming in C, Java, and other
languages. There are three typical cases when a user defines functions which
must return to the runtime during a task. The first case is to receive a response
from a remote server. The second case is to redraw the graphics. The third
case is to prevent the Web browser from freezing in a CPU intensive task. In
these cases, an event-driven programming style is used to return the control
flow to the runtime. However, the event-driven programming style results in
complex program codes even for writing simple procedures. To overcome these
problems, we make it possible for developers to write multithread programs in
ActionScript. In our scheme, first, a programmer writes a multithread program
using our API, and compiles the program into bytecode using the standard
ActionScript compiler. Second, our translator inserts code into each function
in the bytecode program. This code will store the current execution state and
suspend the execution of the function. At runtime, our thread manager calls
the function of the first executable thread. This function executes its bytecode,
and returns to the thread manager when the inserted code is executed. Our
thread manager then calls the function of the next executable thread. With
our scheme, we wrote several programs including animations and downloading
files from servers using the multithread programming style.

1. は じ め に

ActionScriptは，Javaや JavaScriptと同様に，Webブラウザの中で動作するプログラ

ムを開発するための言語である．現在，多くのリッチインターネットアプリケーションが

ActionScriptの技術を利用している．Flashは ActionScriptバイトコードの処理系を内蔵

したWebブラウザのプラグインである．Flashのコンテンツは SWF 1),�1と呼ばれるフォー

マットによってWeb上で配布され，HTMLによって記述されたWebページに埋め込まれ

た形で表示される．開発者が，ActionScriptを用いてユーザプログラムを記述することに

より，インタラクティブ性や豊かな表現力を持ったリッチインターネットアプリケーション

を実現することができる．

Webブラウザの中で動作するプログラムでは，時間のかかる処理が必要になる．たとえば，

遠隔のサーバに対して要求を送信した後，その応答を受け取るまで待つ処理が必要になる．こ

�1 SWF は Small Web Format，あるいは Shockwave Flash の略である．

69 バイトコード変換による ActionScript プログラムのマルチスレッド化

のとき，単純にブロックしてしまうとWebブラウザ全体を停止させてしまう．ActionScript

のプログラムにおいて全体のブロックを避ける場合，開発者はイベント駆動プログラミング

を用いる．ActionScriptのプログラムはその言語のみを習熟した開発者だけでなく，C言

語や Java言語に習熟した開発者によって記述されることがある．C言語や Java言語では，

開発者は全体のブロックを避ける処理をマルチスレッドスタイルで記述する．本論文では

ActionScriptによるプログラムの開発者を次の 2つに分類する．

開発者A 並行処理を記述する場合，マルチスレッドスタイルよりもイベント駆動スタ

イルで容易に記述できる開発者．

開発者 T 並行処理を記述する場合，イベント駆動スタイルよりもマルチスレッドスタ

イルで容易に記述できる開発者．

開発者 Tにとって，ActionScriptがマルチスレッドスタイルのプログラミングを支援し

ていないことは問題となっている2)．本研究で対象とする問題について，詳しくは 2章で述

べる．

このような問題を克服するため，本研究では，C言語や Javaにおいて従来から用いられ

てきたマルチスレッドプログラミングのスタイルを ActionScriptに導入する．これにより，

開発者 Tはイベント駆動スタイルを用いることなく，慣れ親しんだマルチスレッドスタイ

ルを用いて ActionScriptで並行処理を記述することを可能にする．本論文の貢献は，バイ

トコードレベルのコード変換を用いて ActionScript言語におけるマルチスレッドプログラ

ミング環境を実現したことである．提案手法では，開発者は ActionScriptを標準のコンパ

イラでコンパイルしてバイトコードを得る．これに対し，本研究で提供するコード変換プ

ログラムを適用する．コード変換プログラムは，バイトコード中の各関数に実行状態を保

存して処理を中断し，後で再開することを可能とするバイトコードプログラムを挿入する．

ランタイムライブラリはスレッド管理機構とライブラリ関数からなり，開発者が実装したプ

ログラムにリンクして用いる．スレッド管理機構はスレッドを実現するために，コード変換

プログラムによって変換された関数を呼び出す．

提案方式の利点として，まず，ActionScript の文法を変更しないので，既存のコンパイ

ラや開発環境をそのまま利用できることがあげられる．次に，バイトコードのレベルで分岐

命令やレジスタの操作が可能なので，効率良くスレッドのコンテキスト切替えが実現できる

ことがあげられる．また，変換されたコードは手を加えていない標準の ActionScript処理

系で実行できるため，一般ユーザが利用しやすい．

本研究はActionScript 3.0を対象とする．以下，ActionScript 3.0を単純にActionScript

と表記する�1．また，本研究で提案する手法ではマルチスレッドスタイルのプログラミング

を利用するが，一方で開発者 Aにとってはイベント駆動スタイルのプログラミングの方が

自然である．このため，本研究は，開発者 Tのみで構成された 1人以上の開発者のグルー

プを対象とする．開発者のグループの中に開発者 Aが含まれている場合は，本研究の対象

外とする．

本論文の構成を以下に示す．2 章では，ActionScriptにおけるプログラミングがかかえる

制約について説明し，本研究が解決すべき問題を提示する．3 章では，関連研究について述

べる．4 章では，ActionScriptを用いる開発者にマルチスレッドスタイルのプログラミング

を提案し，マルチスレッド化を実現するためのコード変換と実行時ライブラリについて述べ

る．5 章では，今回 ActionScriptで行った本手法の実装について述べる．6 章では，本研究

の有効性を検証する．最後に，7 章では，本論文をまとめ，残された課題について述べる．

2. ActionScriptにおけるイベント駆動プログラミング

ActionScriptの処理系は，外部からの入力を受け取るなど，何かが起きるたびにイベント

を送出する．表 1 に，よく利用されるイベントを示す．ActionScriptのプログラムは，こ

れらのイベントに対応して呼び出されるユーザ定義関数の集合として記述される．このよう

なユーザ定義関数をイベントハンドラと呼ぶ．イベントハンドラを主に用いて行うスタイル

のプログラミングは，イベント駆動プログラミングと呼ばれている．

ActionScriptにおいて並行処理を行うための機構としてイベント駆動が採用された理由

としては，次のようなものがある3)．

• 入出力でブロックを避けるために非同期入出力を用いるため．非同期入出力において，
入出力の完了を知るためにイベントを用いる．

表 1 ActionScript でよく利用されるイベント
Table 1 Frequently used events in ActionScript.

イベント名 送出されるタイミング．
complete 何かの処理が完了した．
click マウスのボタンがクリックされた．
enterFrame 定期的に送出され，画面の再描画が行われる．

�1 ActionScript と呼ばれるプログラム言語には，ほかに主に Flash Player 8 およびそれ以前で使われる Ac-

tionScript 1.0 や ActionScript 2.0 が存在する．ActionScript 2.0 およびそれ以前と ActionScript3.0 と
では，処理系の構造やバイトコードの形式が根本的に異なる．

• デッドロックなどのスレッド間の同期にかかわる複雑性を排し，レビューを簡単にする
ため．

• ActionScriptの処理系の実装を簡潔にし，スレッド実装にともなうオーバヘッドを避

けるため．

ActionScriptではイベントハンドラで時間のかかる処理を続けることは問題がある．Ac-

tionScriptの処理系は，ブラウザの中でシングルスレッドで動作している．1つのイベント

ハンドラを実行しているときに別のイベントが生じたとしても，そのイベントの処理は，実

行中のイベントハンドラがリターンした後に行われる．また，Webブラウザの画面の再描

画も，処理系に制御が戻ってきてから行われる．したがって，イベントハンドラがリターン

しなかった場合，Webブラウザ全体がフリーズしてしまう．

1章で取り上げた開発者 Aは，イベント駆動スタイルを用いて並行処理を行うプログラ

ムを容易に記述することができる．しかし，開発者 Tにとっては，イベント駆動による並

行処理のプログラミングは次に示すような要因からプログラムの可読性が下がり，またプロ

グラミングが難しくなる．

( 1 ) スレッドを用いたプログラミングではブロックできるが，イベント駆動ではブロック

できない．このため素直な実装では，イベント待ちを行う可能性のある部分で関数を

分割する必要がある．たとえば，スレッドを使えば 1つの関数で for文を使ったルー

プを記述しその中でブロックできるが，イベント駆動では複数の関数に分割する必要

がある．また，関数の分割にともない，スレッドでは関数内の局所変数であった変数

がイベント駆動では関数の外の大域変数やクラス変数になってしまう．

( 2 ) スレッドを用いたプログラミングにおいては不要であった，関数をイベントハンドラ

として登録し，登録解除を行う記述が，イベント駆動プログラミングにおいては必要

になる．

これらの問題については，2.1 節，2.2 節，および 2.3 節で，具体的なプログラムの例を

あげて説明する．

2.1 サーバとの通信を行うプログラム

図 1 は ActionScriptを用いてサーバ通信を行うプログラムの実装例である．このプログ

ラムは，ファイルを読み込み，ファイルの読み込みが完了した後にそれを画面に表示する．

Loader オブジェクトは ActionScript に組み込みで用意されたオブジェクトであり，Web

サーバに存在するファイルを読み込む機能を持つ．Loader オブジェクトの load() 関数は，

Webサーバに存在するファイルを読み込む．addChild() 関数は，読み込まれた画像を画

1 var loader:Loader;

2 function run():void {

3 loader = new Loader();

4 var request:URLRequest = new URLRequest("image.gif");

5 loader.load(request); //読み込みを実行する
6 addChild(loader.content); //読み込まれた画像を表示する
7 }

図 1 サーバとの通信を行う不完全なプログラム
Fig. 1 An incomplete program which communicates with a server.

1 var loader:Loader;

2 function run():void {

3 loader = new Loader();

4 var request:URLRequest = new URLRequest("image.gif");

5 loader.contentLoaderInfo.addEventListener("complete", completeHandler);

6 loader.load(request); //読み込みを開始する
7 }

8 function completeHandler(event:Event):void {

9 addChild(loader.content); //読み込まれた画像を表示する
10 }

図 2 イベント駆動によるサーバとの通信を行うプログラム
Fig. 2 A program in event-driven style which communicates with a server.

面に表示する．このプログラムは，開発者 Aにとっては難しいが，開発者 Tにとっては分

かりやすい．

図 1 のプログラムは，実際には期待された動作を行わない．理由は，load() 関数の動作

にある．サーバからのファイルの読み込みには時間がかかる．単純にその完了を待っていた

のでは，Web ブラウザ全体がフリーズしてしまう．これを避けるために，load() 関数は

サーバに要求を送信すると応答の受信を待たずにリターンする．load() 関数を呼び出した

後，ファイルの読み込みが完了すると処理系は complete イベントを送出する．ファイルの

読み込みの完了を待って画像の表示を行うためには，complete イベントに対するイベント

ハンドラを登録する必要がある．

図 2 は，図 1 のプログラムにイベントハンドラを追加したプログラムである．

completeHandler() 関数は，プログラムに追加したイベントハンドラである．このイベ

ントハンドラを Loader によるファイル読み込みが完了した後に呼び出されるように登録す

るために，組み込みの addEventListener() 関数を用いている．このプログラムは，開発

1 function run():void {

2 for (var i:int = 0; i < 100; i++) {

3 this.x += 5; // 表示オブジェクトを右に 5ピクセル動かす
4 redraw(); //再描画を行う
5 }

6 }

図 3 for 文によるアニメーションプログラム
Fig. 3 A program which displays an animation using the for statement.

1 function run():void {

2 addEventListener("enterFrame", enterFrameHandler);

3 }

4 var i:int = 0;

5 function enterFrameHandler(event:Event):void {

6 this.x += 5; // 表示オブジェクトを右に 5ピクセル動かす
7 if (++i >= 100)

8 removeEventListener("enterFrame", enterFrameHandler);

9 }

図 4 イベント駆動によるアニメーションプログラム
Fig. 4 A program which displays an animation in an event-driven style.

者 Aにとっては分かりやすいが，開発者 Tにとっては図 1 のプログラムよりも分かりにく

くなっている．

この例では，イベント駆動によるプログラムに書き換えたことにより，1つの処理を実装

するために 2つの関数を定義する必要が生じている．

2.2 アニメーションを行うプログラム

図 3 は for文で記述されたアニメーションプログラムの実装例である．このプログラム

は，表示オブジェクトを右に 5ピクセルずつ，100回動かしている．また，redraw() 関数

は画面の再描画を行った後，次に画面の再描画が必要になるまで待つ関数である．しかし，

実際の ActionScriptには redraw() のような関数は存在しない．このプログラムは，開発

者 Aにとっては難しいが，開発者 Tにとっては分かりやすい．

実際の ActionScriptでアニメーションプログラムを実装するためには，イベント駆動プ

ログラミングのスタイルを用いて，図 4 に示すようなプログラムを記述する必要がある．こ

の例では，enterFrameHandler() 関数が enterFrame イベントの発生により呼び出され

る．この際，プログラム中の表示オブジェクトが右に 5ピクセルずつ動く．このプログラム

は，開発者 Aにとっては分かりやすいが，開発者 Tにとっては図 3 のプログラムよりも分

1 function run():void {

2 for (var i:int = 0; i < 10000; i++) {

3 heavyTask();

4 }

5 }

6 function heavyTask():void {

7 for (var i:int = 0; i < 10000; i++) {

8 //重い処理
9 }

10 }

図 5 CPU 時間を消費する処理を行うプログラム
Fig. 5 A program which consumes CPU time.

1 function run():void {

2 addEventListener("enterFrame", enterFrameHandler);

3 }

4 var i:int = 0;

5 function enterFrameHandler(event:Event):void {

6 heavyTask();

7 if (++i >= 10000)

8 removeEventListener("enterFrame", enterFrameHandler);

9 }

10 function heavyTask():void {

11 for (var i:int = 0; i < 10000; i++) {

12 //重い処理
13 }

14 }

図 6 イベント駆動による CPU 時間を消費する処理を行うプログラム
Fig. 6 A program which consumes CPU time in an event-driven style.

かりにくくなっている．

この例では，イベント駆動によるプログラムに書き換えたことにより，1つの処理を実装

するために 2つの関数を定義する必要が生じている．ここでも，2.1 節の例と同じような問

題が発生している．さらに，図 3 のように繰返し構造を for文を用いて記述することはでき

なくなっているという問題も発生している．

2.3 CPU時間を消費するプログラム

図 5 は CPU時間を消費する処理を行う heavyTask() 関数を 10,000回繰り返して呼び

出すプログラムである．このプログラムは，開発者 A にとっては難しいが，開発者 T に

とっては分かりやすい．このプログラムをWebブラウザ上で実行すると，すべての計算が

情報処理学会論文誌 プログラミング Vol. 4 No. 3 68–83 (June 2011) c© 2011 Information Processing Society of Japan

終わるまでの間Webブラウザがフリーズしてしまう．Webブラウザのフリーズを避けるた

めには，図 6 に示すような，イベントハンドラを用いたプログラムを記述する必要がある．

図 6 のプログラムでは，heavyTask() 関数を 1回呼ぶごとにイベントハンドラからリター

ンしている．このプログラムは，開発者 Aにとっては分かりやすいが，開発者 Tにとって

は図 5 のプログラムよりも分かりにくくなっている．

ここでも，2.2 節に示した問題と同様の問題が生じている．

3. 関 連 研 究

ライブラリによりActionScriptでイベントハンドラの登録・解除の煩わしさを軽減する手

法がいくつか提案されている．PseudoThreadは，2.3節で述べたようなループを含む CPU

時間を消費するプログラムを簡単に記述するためのライブラリである4)．PseudoThreadラ

イブラリを利用する開発者は，ループ処理の 1回分の処理を．関数として定義する．Pseu-

doThreadライブラリは，この関数を定期的に呼び出す．ActionScript Thread Library（そ

うめん）は，PseudoThreadよりも汎用性があり，ループ以外の処理も記述することができ

る2)．そうめんライブラリを利用する開発者は，1つの処理の途中で中断したい場合，その

処理を複数の関数に手作業で分割して記述する．各関数は，末尾で次に実行すべき関数をそ

うめんライブラリに返す．そうめんライブラリは，指定された関数を実行したり，別の制御

の流れに属する関数を実行したり，あるいは，ActionScript処理系に制御を戻したりする．

本研究の手法では，これらの手法とは異なり，開発者は 1つの処理をライブラリ側の都合で

複数の関数に分割する必要はない．また，バイトコードレベルのコード変換を用いている点

が異なる．

JavaScript言語は，Webブラウザの上で動作する処理系である．文献 5)では，JavaScript

におけるイベント駆動プログラミングスタイルに関連して同期処理を非同期処理で記述しな

ければならないという問題があることを指摘している．ActionScriptにも類似の問題があ

る．JavaScriptのプログラムに対してコード変換を行い，JavaScriptによるマルチスレッ

ドプログラミングを実現する試みとしては，研究 6)やライブラリ 7)がある．JavaScript

ではプログラムのソースコードをプログラムから取得することが容易である．そのため，こ

れらのシステムでは JavaScriptにおけるコード変換を実行時にソースコードレベルで行う．

これに対して，本研究では ActionScriptのプログラムを実行前にあらかじめバイトコード

コードレベルで変換することによりマルチスレッドプログラミングを実現する手法を提案

する．

Web Workers 8) は，JavaScriptの処理系自体を改良し，マルチスレッドプログラミング

を可能とする技術である．Web Workersにより提供されるスレッドは本研究や研究 6)など

が提供するコルーチンに基づくマルチスレッドではなく，OSレベルのプリエンプティブな

スレッドである．このため，Web Workersによるスレッドは低オーバヘッドであり，マル

チコアプロセッサの恩恵を受ける．一方，それぞれのWeb Workersスレッドはオブジェク

トを共有しておらず，スレッド間の通信は文字列で行う必要がある．他にも，document オ

ブジェクトの操作を行うことができるスレッドを新たに生成することはできないなど，Web

Workers が提供するスレッドにはいくつかの制限がある．本研究の手法が提供するマルチ

スレッドはコルーチンに基づくマルチスレッドであるが，すべてのスレッドがすべてのオブ

ジェクトにアクセス可能である．

Javaのバイトコードを変換することによってプログラムの機能を変更する試みとしては

Javassist 9) が存在する．本研究では ActionScriptのバイトコードを変換することによって

プログラムの機能を変更する．バイトコード変換を実現するため，本研究では swfassist 10)

と Tamarin 11) の ABC（ActionScript Byte Code）アセンブラと逆アセンブラを組み合わ

せて用いる．Javassist を用いたコード変換を制御するためには，Java の機能であるアノ

テーションを用いることができる．本研究においても，コード変換の対象となるプログラム

を指定するためにアノテーションを用いる．本研究のコード変換は ActionScriptを対象と

する．

C言語や C++言語で書かれたプログラムを ActionScriptバイトコードにコンパイルす

る技術として，Adobe Alchemy 12) が存在する．Alchemyは LLVM 13) を利用することに

より，Webブラウザをフリーズさせることなく実行することができる ActionScriptのバイ

トコードプログラムを生成する．本研究と Alchemyは，いずれも ActionScriptの処理系の

上に何らかのプログラムを動かすための機構を設ける．しかし，Alchemyは C言語や C++

言語で書かれたプログラムを対象とする．本研究は ActionScriptで記述されたプログラム

を対象とする．

Adobeが Flash Playerの一部として提供している PixelBender 14) は，科学技術計算や

画像処理など，CPUに負荷がかかる処理を Flash Player上で並列に行うための仕組みであ

る．PixelBenderのプログラムはActionScriptとは異なる言語で記述される．また，専用の

開発環境が用意されている．一方，PixelBenderではActionScriptのオブジェクトにアクセ

スすることができず，できることは数値演算などに限られる．このことから，PixelBender

は 1章で述べた開発者 Tがマルチスレッドスタイルのプログラムを記述するために利用す

ることはできない．本研究の手法は，ActionScriptを用いたマルチスレッドスタイルのプ

ログラミングを行うためのものである．

文献 15)では，C言語におけるユーザレベルのスレッド・ライブラリの実現に継続を利

用している．また，様々なプログラミング言語でスレッドの実装に必要なコードをコード変

換の過程で挿入する方法がすでに提案されている16)–19)．本研究では，ActionScript言語に

おけるスレッドの実装にバイトコード変換を用いる方法を提案する．

4. 提 案 手 法

4.1 概要と構成

ActionScriptのイベント駆動プログラミングは，1章で取り上げた開発者 Tにとっては

2 章で説明したような問題をかかえている．一方で，C言語や Javaでは，ライブラリや処

理系が提供するスレッド APIを用いてマルチスレッドプログラミングを行うことが可能で

ある．マルチスレッドプログラミングでは，時間のかかる処理を別のスレッドで動かすこと

ができる．ActionScriptでスレッドを使うことが可能ならば，開発者 Tは 2 章で述べたよ

うな簡単な記述で時間のかかる処理を行うプログラムを実装することが可能になる．

本研究は，ActionScriptにおいて時間のかかる処理をイベントハンドラを用いずに逐次

的に記述できるようにすることを目的とする．目的を達成する手段として，本研究では，

ActionScriptにスレッドを導入して，マルチスレッドプログラミング環境を実現すること

を提案する．

図 7 は，ActionScriptを用いる開発者が提案手法を利用して開発を行うワークフローで

ある．まず，開発者は提案手法が提供する APIを利用し，マルチスレッドプログラミング

によってプログラムを記述する．次に，開発者は通常の ActionScriptコンパイラを利用し，

ソースコードから中間 SWFファイルを得る．最後に，開発者はこの中間 SWFファイルを

提案手法が提供するコード変換プログラムによってコード変換し，変換後の SWF ファイ

ルを得る．変換後の SWFファイルがWebブラウザで開かれると，SWFファイルに組み

込まれたランタイムライブラリはコード変換プログラムによって変換されたプログラムを

マルチスレッド的に実行する．提案手法における開発者が記述するプログラムについては，

4.2 節で説明する．

本手法では，バイトコードレベルの変換を採用した．これには以下の 3つの利点がある．

1 つ目の利点は，本手法では，ActionScript の文法には変更を加えないことである．この

ため，開発者は新たなプログラミング言語を覚える必要がない．また，開発者は既存の開

図 7 提案手法のワークフロー
Fig. 7 The workflow of the proposed method.

発環境をマルチスレッドプログラミングにおいてもそのまま利用できる．さらに，標準の

ActionScriptコンパイラによってコンパイル時に型エラーをチェックすることができる．2つ

目の利点は，ActionScriptの処理系に変更を加えないことである．このため，既存の Flash

プラグインでマルチスレッドの ActionScriptアプリケーションを利用できる．3つ目の利

点は，コード変換の際に ActionScriptから直接扱うことができないバイトコード命令を利

用できることである．たとえば，関数内の任意の位置へのジャンプを行うバイトコード命令

である jump は，対応する ActionScriptの言語要素が存在しない．このため，ソースコー

ドレベルの変換では，関数内の任意の位置に直接ジャンプすることができない．

本手法では，ActionScriptプログラムに対してコード変換を行い，コルーチンに基づく

マルチスレッドを実現する．ネイティブにスレッドをサポートしない ActionScript処理系

においてスレッド間のコンテキスト切替えを可能とするために，本手法では以下の 4つの

要素を用いる．

中断可能関数 処理を中断して後で再開することができる関数である．

継続オブジェクト 中断可能関数の実行状態を保存するオブジェクトである．

スレッドオブジェクト スレッドを表現するオブジェクトである．

駆動プログラム 現在実行中のスレッドオブジェクトを管理するプログラムである．

このうち，中断可能関数は開発者が実装したプログラムをコンパイルして生成されたバイ

トコードを入力として，本手法において提供されるコード変換プログラムによって生成され

る．残りの 3つ，すなわち継続オブジェクト，スレッドオブジェクト，駆動プログラムは，

本手法において提供されるランタイムライブラリの一部である．

図 8 に，変換後の SWFファイルがWebブラウザ内で実行されている様子を示す．駆動

図 8 提案手法の構成
Fig. 8 The structure of the proposed method.

プログラムは実行可能なスレッドオブジェクトを選び，制御を渡す．スレッドオブジェクト

は，制御を受け取ると保持している継続オブジェクトから中断可能関数を取り出して，それ

を呼び出す．中断可能関数は開発者が記述したとおりの処理を行うが，コード変換により挿

入されたコードを実行すると，現在の関数の実行環境を継続オブジェクトに保存して，制

御をスレッドオブジェクトに戻す．スレッドオブジェクトはこの継続オブジェクトを保存し

て，制御を駆動プログラムに返す．駆動プログラムは，ここで次の実行可能なスレッドオブ

ジェクトに制御を渡す．

中断可能関数と継続オブジェクトの実現については，4.3 節で説明する．駆動プログラム

とスレッドオブジェクトについては，4.4 節で説明する．

4.2 マルチスレッドプログラミング

開発者は以下の手順を踏み，提案手法を利用したマルチスレッド ActionScriptプログラ

ムを開発する．

提案手法では，スレッドとして実行可能にするためにバイトコード変換の対象となる関数

にのみ [Suspendable]というアノテーションを付加する．アノテーションが付加されていな

い関数は，変換せずにそのまま出力する．特定の関数の並行処理のためのマークを付加す

ることは，Cilk-5 言語20) など，多くのプログラミング言語で用いられている手法である．

表 2 関数呼び出しの可否
Table 2 Semantics of function calling.

呼び出し先
通常の関数 中断可能関数

呼び出し元 通常の関数 可 不可
中断可能関数 可 可

1 new Thread(func, [arg1, arg2, ...]).start();

図 9 新しいスレッドを生成するプログラム
Fig. 9 A program to create a new thread.

また，Java言語21) においても，意味は異なるが synchronizedという構文を導入している．

すべての関数をバイトコード変換の対象とすることも考えられるが，本研究では特定の関数

だけを変換の対象とした．その理由は，CPU時間を消費する処理において，通常の関数と

中断可能関数を組み合わせることで，オーバヘッドをともなう中断にともなう処理の頻度を

調整しやすくするためである．この調整について詳しくは，6.2節で述べる．

中断可能関数と通常の関数は関数の呼び出しに関して異なる性質を持つ（表 2）．中断可

能関数と通常の関数は，直接呼び出すことができる関数が異なる．通常の関数は，通常の関

数のみを直接呼び出すことができる．一方，中断可能関数は，通常の関数に加えて中断可能

関数を呼び出すことができる．

中断可能関数を通常の関数から利用するためには，スレッドを生成する．スレッドを生成

するためには，スレッドオブジェクトを作成したうえで中断可能関数とそれに渡す引数を指

定し，スレッドオブジェクトの start() メソッドを呼び出す．この様子を図 9 に示す．ス

レッドの生成は，通常の関数と中断可能関数の両方から行うことができる．

通常の ActionScriptと異なり，中断可能関数内では for文によるループの途中で定期的

に制御が ActionScriptの処理系に戻る．このため，開発者は CPU時間を消費するループ

を細切れに分割することなく，for文を利用して素直に記述することができる．逆に通常の

関数が実行中の間，制御が処理系に戻ることはない．この性質を利用することでアトミック

な処理を実現することができる．たとえば，アニメーションにおいて 1つの図形をアトミッ

クに作成したい場合，その処理を通常の関数として記述して，中断可能関数から呼び出せば

よい．

本研究では，マルチスレッドプログラミングを行うことを助けるためのいくつかのライブ

表 3 提案機構が提供するライブラリ関数
Table 3 Library methods offered by proposed method.

関数名 関数の説明
yield() 制御を処理系に戻す．
redraw() 画面を再描画するまでスレッドを一時停止する．
sleep() 指定した時間が経過するまでスレッドを一時停止する．
wait() 指定したイベントが送出されるまでスレッドを一時停止する．

1 [Suspendable]

2 function run():void {

3 var loader:Loader = new Loader();

4 var request:URLRequest = new URLRequest("image.gif");

5 loader.load(request); //読み込みを開始する
6 wait(loader, Event.COMPLETE);

7 addChild(loader.content); //読み込まれた画像を表示する
8 }

図 10 提案手法を用いたサーバとの通信を行うプログラム
Fig. 10 A program using the proposed method which communicates with a server.

ラリ関数を提供する．本研究で実装したライブラリ関数を表 3 に示す．これらのライブラ

リ関数は中断可能関数から呼び出すことができる．ライブラリ関数 wait()は，入出力の完

了などの標準のイベントのほかに，ユーザ定義のイベントも待つことができる．この機能を

利用することで，C言語（Pthreads）や Java言語における条件変数を使った待ち合わせと

似た処理を記述することができる．たとえば，ある条件が満たされるまで待ちたいスレッド

がユーザ定義イベントを引数にして wait()を呼び出し，別のスレッドがそのユーザ定義イ

ベントを発生させ待っているスレッドを起こすことができる．

以上に述べた相違点を除いて，開発者の目から見た中断可能関数と通常の関数は同じであ

る．たとえば，開発者は通常の関数と同じ ActionScriptの文法を用いて中断可能関数を記

述する．また，通常の関数と中断可能関数はともに ActionScriptの処理系が提供する組み

込みの機能を利用することができる．

中断可能関数を実装することによって 2 章で述べた問題を解決できることを，4.2.1 項，

4.2.2 項，および 4.2.3 項で示す．

4.2.1 サーバとの通信を行う中断可能関数

図 10 は，2.1 節で例示した処理と同一の処理を行う中断可能関数である．この例では，

1 [Suspendable]

2 function run():void {

3 for (var i:int = 0; i < 100; i++) {

4 this.x += 5; // 表示オブジェクトを右に 5ピクセル動かす
5 redraw(); //再描画を行う
6 }

7 }

図 11 提案手法を用いたアニメーションプログラム
Fig. 11 A program which displays an animation using the proposed method.

1 [Suspendable]

2 function run():void {

3 for (var i:int = 0; i < 10000; i++) {

4 heavyTask();

5 }

6 }

7 [Suspendable]

8 function heavyTask():void {

9 for (var i:int = 0; i < 10000; i++) {

10 //重い処理
11 }

12 }

図 12 提案手法を用いた CPU 時間を消費する処理を行うプログラム
Fig. 12 A program which consumes cpu time using the proposed method.

図 2 における addEventListener() 関数の呼び出しを wait() 関数で置き換えることによ

り，処理の流れを 1つの関数にまとめて記述することができている．

4.2.2 アニメーションを行う中断可能関数

図 11 は，2.2 節で例示した処理と同一の処理を行う中断可能関数である．この例は，図 3

のプログラムに [Suspendable] の記述を付加したものである．この例では，処理の流れを

1つの関数にまとめて記述することができているうえ，さらに繰返し構造を for文を用いて

記述することができている．

4.2.3 CPU時間を消費する処理を行う中断可能関数

図 12 は，2.3 節で例示した処理と同一の処理を行う中断可能関数である．この例は，図 5

のプログラムに [Suspendable] アノテーションの記述を付加したものである．この例では，

Webブラウザをフリーズさせることなく，また処理を細かく分割することもなく CPU時

間を消費する処理を行うことが可能となっている．

4.3 中断可能関数と継続オブジェクト

本研究では，バイトコードレベルのコード変換によって中断可能関数を生成する．コード

変換では，関数の命令列中に継続オブジェクトを返す命令列を挿入する．ここで挿入される

命令列を，本論文では中断コードと呼ぶことにする．中断コードは，中断可能関数の現在の

実行状態を継続オブジェクトに保存してリターンするためのコードと中断可能関数が継続オ

ブジェクトから実行状態を読み込んで再開するためのコードから構成される．

中断可能関数の性質については，4.3.1項で述べる．継続オブジェクトについては，4.3.2項

で述べる．中断コードについては，4.3.3 項で述べる．

4.3.1 中断可能関数

中断可能関数は，次の 2つのうち 1つを引数としてとる．

( 1 ) 本来の引数．

( 2 ) 継続オブジェクト．

中断可能関数を最初に呼び出すときは，本来の引数を渡す．中断可能関数の実行を再開す

るときは，継続オブジェクトを渡す．中断可能関数は，このどちらを受け取ったかを判断す

るために，引数の型を調べる．

中断可能関数は，次の 2つのうち 1つを返値として返す．

( 1 ) 本来の返値．

( 2 ) 継続オブジェクト．

中断可能関数の実行が完了したときは，呼び出し元は本来の返値を受け取る．中断可能関

数の実行が中断したときは，呼び出し元は継続オブジェクトを受け取る．中断可能関数の呼

び出し元は，このどちらを受け取ったかを判断するために，受け取った返値の型を調べる．

4.3.2 継続オブジェクト

継続オブジェクトには以下の 4つの値が格納される．

( 1 ) 中断可能関数への参照．

( 2 ) 中断可能関数の実行を再開する地点．

( 3 ) 中断可能関数の実行状態．

( 4 ) 次に実行する処理を表現する継続オブジェクトへの参照．

図 13 のようなプログラムを実行すると，複数の継続オブジェクトが生成され，それらが

1つのリストにつながれる．この様子を図 14 に示す．この場合，継続オブジェクトは 2つ

作成される．関数 g() は最初に作成された継続オブジェクトに実行状態を保存し，それを関

数 f() に返す．関数 f() は別の継続オブジェクトを作成して実行状態を保存し，関数 g()

1 [Suspendable]

2 function f():int {

3 return g();

4 }

5 [Suspendable]

6 function g():int {

7 yield(); //ここで中断処理が行われる
8 return 1;

9 }

図 13 中断可能関数から他の中断可能関数を呼び出すプログラム
Fig. 13 A program which calls a suspendable function from another suspendable function.

図 14 複数の中断可能関数の中断
Fig. 14 Suspending multiple suspendable functions.

から受け取った継続オブジェクトの次に実行する処理として登録する．関数 f() は関数 g()

から受け取った継続オブジェクトを返値としてスレッドオブジェクトに返す．スレッドオブ

ジェクトがこの実行状態を再開するときは，まず関数 g() の継続オブジェクトの実行状態

を復元する．関数 f() の実行状態の復元は，関数 g() の実行が完了するまで遅延される．

4.3.3 中断コード

本研究では，3種類の中断コードを用いた．

1つ目は，中断可能関数内に記述された yield() 関数の呼び出しから変換される中断コー

ドである．この中断コードでは，必ず中断処理が行われる．これは，本研究で提供するライ

ブラリ関数のうち，redraw() 関数や wait() 関数を実装するために用いた．これらのライ

ブラリ関数では，スレッドを待ち状態に移行したうえで yield() の処理を行う．

2つ目は，中断可能関数内で関数呼び出しが行われている場所に挿入される中断コードで

ある．この中断コードでは，呼び出された中断可能関数が継続オブジェクトを返した場合

に，中断処理を行う．呼び出された関数が返値として継続オブジェクトではない値を返した

場合は，そのまま処理を続ける．

3つ目は，中断可能関数内のループ中に挿入される中断コードである．これは，for文な

どによって中断可能関数内で CPU時間を消費するループを実装した際に，ループの途中で

いったん中断可能関数からリターンするためのものである．この中断コードについては，通

過した回数をグローバル変数を用いて数え，グローバル変数が一定の値に達するごとに中断

処理を行う．これは，短いループにおいて 1回の繰返しごとに中断処理を行うことは効率が

悪いためである．現在の実装では，この値として標準で 100を用いている．

コード変換によって埋め込まれる中断コードについて述べる．実際のコード変換は Ac-

tionScript のバイトコードに対して行われる．ActionScript のバイトコードを直接示すの

は煩雑である．代わりに ActionScript風の擬似コードを用いる．

図 15 は，図 11 のプログラムを変換して生成された中断可能関数である．

ソース中の Continuation は継続オブジェクトの型であり，c は変換処理によって追加

された引数である（図 15 の 1行目）．関数の先頭には c が継続オブジェクトであった場合

に継続オブジェクトから状態を再開するコードが挿入される（図 15 の 2–6行目）．

ライブラリ関数 redraw() 関数は中断可能関数である．このため，redraw() 関数の返

値の型を調べ，それが継続オブジェクトであれば中断処理を行うコードが挿入されている

（図 15 の 9–16行目）．ここで中断処理が行われると，4.3.2 項で述べたように，redraw()

関数が生成する継続オブジェクトの後にこの中断可能関数が生成する継続オブジェクトがつ

ながれる（図 15 の 13行目の queue() 関数）．この地点で中断処理が行われた場合，処理

はこの中断コードの直後（図 15 の 15行目）から再開される．

元の関数でリターンが行われていた場所では，呼び出し元に制御を移すためにコード変換

を行う（図 15 の 24–31行目）．ここでは，この中断可能関数が直接呼び出されたのか，あ

るいはスレッドオブジェクトから呼び出されたのかを調べる（図 15 の 24行目）．この中断

可能関数がスレッドオブジェクトから呼び出されていた場合，継続オブジェクトにつながれ

た続きの継続オブジェクトに返値を渡し，実行を再開する（図 15 の 26–27行目）．続きの

継続オブジェクトは，図 15 の 4行目のような処理を行い，渡された返値を受け取る．この

中断可能関数が呼び出し元の関数から直接呼び出された場合は，返値を直接リターンするこ

とにより呼び出し元の関数に返値を渡す（図 15 の 32行目）．

本手法では，関数呼び出しを行う際に，それが通常の関数であるか中断可能関数である

1 function run(c:Continuation = null):Continuation {

2 if (c) {

3 i = c.locals[1];

4 c1 = c.returnValue;

5 goto c.label;

6 }

7 for (var i:int = 0; i < 100; i++) {

8 this.x += 5; // 表示オブジェクトを右に 5ピクセル動かす
9 var c1:* = redraw();

10 if (c1 is Continuation) {

11 if (c == null) c = new Continuation();

12 c.func = this.run; c.label = $L1; c.locals[1] = i;

13 c1.queue(c);

14 return c1;

15 $L1:

16 }

17 if ( ++loop_counter >= max_loop_counter ) {

18 if (c == null) c = new Continuation();

19 c.func = this.run; c.label = $L2; c.locals[1] = i;

20 return c;

21 $L2:

22 }

23 }

24 if (c) {

25 if (c.next) {

26 c.next.returnValue = null;

27 return c.next;

28 } else {

29 return null;

30 }

31 }

32 return null;

33 }

図 15 コード変換後のプログラム
Fig. 15 Pseudo code of a program after code translation.

かを区別しない．そのため，中断可能関数の中ではすべての関数呼び出しについて，この

ような中断処理を行うコードを挿入する．また，ループの末尾にも中断コードを挿入する

（図 15 の 17–22行目）．このコードは，ループを通過した回数を数え，それが一定の値に達

するごとに中断処理を行う．

4.4 駆動プログラムとスレッドオブジェクト

本手法において提供される駆動プログラムはすべてのスレッドオブジェクトを管理する．

スレッドオブジェクトは start() メソッドが呼ばれると自身を駆動プログラムに登録し

て実行可能状態に移行する．実行可能状態のスレッドオブジェクトは駆動プログラムから制

御を受け取るたび，自身に登録された継続オブジェクトから中断可能関数を取り出し，それ

に継続オブジェクトを渡す．継続オブジェクトを渡された中断可能関数は実行を再開する．

中断可能関数が継続オブジェクトを返した場合，スレッドオブジェクトはそれを保持し，制

御を駆動プログラムに渡す．中断可能関数が継続オブジェクト以外の型を持つ値を返した場

合，スレッドは動作を終了する．

スレッドは，ライブラリ関数 redraw()，sleep()，wait() が呼び出されると，待ち状

態に移行する．待ち状態のスレッドは，イベントハンドラで対応するイベントを受信したと

きに実行可能状態にする．

5. 実 装

5.1 ActionScript Byte CodeとActionScript Virtual Machine

ABCのプログラムは SWFに格納されている．ABCを実行することができる処理系は，

ActionScript Virtual Machine，または縮めてAVMと呼ばれている．AVMはFlash Player

の一部である．

AVMは ABCのプログラムを読み込むと，バイトコードを実行する．AVMにおける関

数フレームには以下の 3つのメモリ領域が存在する．

引数領域 引数を格納する．

スコープ領域 名前解決を行う対象のオブジェクトを格納する．

ローカルレジスタ バイトコードプログラムが自由に利用することができる．

これらの領域の大きさは関数ごとに異なる．これらの値は ABCの中に記述されている．

また，AVMはバイトコードプログラムの実行に先立ち，ABCデータが範囲外のメモリ

をアクセスしたり，不適切な型変換を行ったりしてしまうなどの危険な挙動を行わないよう，

ABCデータに対してコード検証を行う機能を持つ．AVMはコード検証の際，バイトコー

ドに含まれる命令列を抽象的に実行し，起こりうる実行の流れをシミュレートする．コード

検証では，引数スタックやスコープスタックの深さが深くなりすぎないことを，また引数ス

タックやローカル変数に含まれる値の型が矛盾を起こさないことをチェックする．

5.2 ABCのコード変換

本研究では，SWFファイルに含まれるデータのうち，ABCについてコード変換を行い，

それ以外のデータをそのまま保持する．今回の実装では，これを ActionScriptのプログラ

ムとして実現した．SWFおよび ABCはバイナリフォーマットであるため，直接コード変

換を行う ActionScriptのプログラムを実装することは困難である．このため，本手法では

図 16 コード変換プログラムの構造
Fig. 16 The structure of the code translator.

SWFファイルから ABCのバイナリを抽出したうえで，ABCを逆アセンブルし，得られた

アセンブリ言語表現に対してコード変換を行う．本手法のコード変換プログラムは，ABC

に含まれる関数のうち，[Suspendable] アノテーションの付加されたものを変換する．最

後に，アセンブリ言語表現をアセンブルし，得られた ABCのバイナリを SWFファイルに

埋め込む．

ActionScriptからローカルのファイルシステムに対する読み書きを行うためには，AIR

ランタイム22) を用いた．本手法において用いるコード変換プログラムの実装にあたって

は，図 16 に示すソフトウェアを利用した．SWFファイルから ABCによるバイトコード

プログラムを抽出し，また変換されたバイトコードプログラムを再度 SWF ファイル中に

埋め込む処理は，swfassist 10) を利用して行った．アセンブラと逆アセンブラについては，

Tamarin 11) に付属する abcenc プログラムと abcdec プログラムを用いた．abcenc プロ

グラムが用いるアセンブリ言語は JSONフォーマットを用いている．これを ActionScript

のオブジェクトとして読み込み，プログラムから操作するために，as3corelib 23)内の JSON

パーサに手を加えて利用した．

1 class Continuation {

2 var func:Function;

3 var label:int;

4

5 var operands:Array;

6 var scopes:Array;

7 var locals:Array;

8 var returnValue:*;

9

10 var next:Continuation = null;

11

12 function queue(c:Continuation):Continuation;

13 function resume():*;

14 function resumeWithReturnValue(value:*):*;

15 }

図 17 継続オブジェクト
Fig. 17 Continuation Object.

5.3 継続オブジェクトの実装

4.3.2 項で述べた継続オブジェクトの ActionScriptにおける実装を図 17 に示す．最初の

6つの変数は，中断可能関数の実行状態を保存するための変数である．最後の 3つの関数は

継続オブジェクトを使いやすくするためのユーティリティ関数である．

中断可能関数への参照は funcに保存される．中断可能関数の実行を再開する位置は label

に保存される．label は中断可能関数内の命令のメモリアドレスを間接的に表現する整数値

である．継続オブジェクトから関数の実行状態を復元する際は，まず実行関数を呼び出し，

次に label の値に応じて最後に処理を中断した中断コードの直後に制御を移す．図 17 の，

operands，scopes，および locals は，関数フレームに含まれる引数領域，スコープ領域，

ローカルレジスタの 3つを保存するための ActionScriptの配列である．図 17 に存在する

next は，4.3.3 項で述べた，継続オブジェクトのリストを表すためのフィールドである．

継続オブジェクトのユーティリティ関数について述べる．queue() 関数は中断可能関数

の実行が中断されるときに呼び出され，next フィールドによって形成されるリンクリスト

の末尾に継続オブジェクトを追加する．resume() 関数と resumeWithReturnValue() 関

数は中断可能関数の実行を再開する．このうち，resume() 関数はスレッドオブジェクトか

ら呼び出される．resumeWithReturnValue() 関数は 4.3.3 項で述べたような場合に用いら

れる．

5.1 節で述べたとおり，AVMは ABC読み込み時にコード検証を行い，関数フレームに

格納された値の静的な型情報に一貫性があることを保証する．これは配列から関数フレーム

に値を書き戻す際に問題になる．ActionScriptの配列には任意の型の値を格納することが

できるため，AVMはコードの検証に失敗する．この問題を防ぐため，今回行った実装では

オブジェクトを * 型にキャストするバイトコード命令を挿入した．この処置によって関数

の行う処理は変化しないが，実行時の型検査によるオーバヘッドが発生する．

6. 評 価

6.1 記 述 性

4.2 節では，本手法を用いることにより，従来の ActionScriptにおいてイベント駆動に

よって記述する必要があった処理をマルチスレッドスタイルで記述することが可能であるこ

とを示した．マルチスレッドプログラミングには，ActionScriptにおけるイベント駆動プ

ログラミングには存在しなかった，以下の利点がある．

( 1 ) 本手法でスレッドを用いたプログラミングでは関数の途中でブロックできる．特に

for文を使ったループを記述しその中でブロックすることができることは，記述性を

大きく高める．イベント駆動では，イベント待ちのために 1つの処理を複数の関数に

分割しなければならないことがあったが，本手法ではそのための分割は不要である．

関数を分割する必要がないので，関数内の局所変数が使いやすい．

( 2 ) イベントハンドラを登録・解除する処理を削減することができる．

本手法では，通常の関数から中断可能関数（[Suspendable]アノテーションが付加されて

いる）を呼び出すことはできない．この呼び出しを自動的に検出することは容易ではなく，

現在実装されていない．したがって，コードレビューにおいてこの点を確認する必要があ

り，コードレビューの手間が増大する．また，他のマルチスレッドを採用した言語と同様に，

デッドロックなどのスレッド間の同期に関わる問題にも直面するすることになる．

6.2 オーバヘッド

本手法を利用することによるオーバヘッドは，その発生源によって 2つを考えることがで

きる．1つは，本手法のコード変換プログラムによりコードが変更されることによって生じ

るオーバヘッドである．もう 1つは，スレッド管理機構を駆動することによって生じるオー

バヘッドである．

これらのオーバヘッドを計測するために，本研究ではループや関数呼び出しを含む Ac-

tionScriptプログラムを通常の関数と中断可能関数を用いて実装し，実行時間を測定した．

実行時間の測定に用いた環境を，表 4 に示す．

表 4 実行時間の測定に用いた環境
Table 4 The environment for the benchmark test.

OS Windows 7 Professional 32 bit

プロセッサ Intel Core i7 2.67GHz

メモリ 6GB

Flash Player バージョン 10.1.102.64

表 5 提案機構のオーバヘッド
Table 5 Overhead of proposed method.

操作 1 回あたりの実行時間
通常の関数 中断可能関数

for 文による 1 ループ 2.3 ns 11.4 ns

通常の関数の呼び出し 3.8 ns 21.9 ns

中断可能関数の呼び出し – 35.6 ns

コンテキスト切替え – 252.7 ns

通常の関数と中断可能関数の実行時間の差から，提案機構が持つオーバヘッドを計算し

た．その結果を表 5 に示す．

まず，for文によるループを中断可能関数に変換すると実行時間が 1ループあたり 9.1ナ

ノ秒増加する．これは，今回の実装で，5.3 節で述べたように，コード中にキャストを行う

処理を挿入したためである．次に，通常の関数における空の関数呼び出しの実行時間は 1回

あたり 3.8ナノ秒であるのに対し，中断可能関数における空の通常の関数呼び出しの実行時

間は 1回あたり 21.9ナノ秒と，通常の関数からの関数呼び出しと比べて 18.1ナノ秒増加し

た．これは，4.3.3 項で述べたように，中断可能関数が他の関数を呼び出す際には，返値の

型を調べて条件分岐を行う必要があるためである．中断可能関数における空の中断可能関数

呼び出しの実行時間は 35.6ナノ秒であり，中断可能関数から通常の関数を呼び出した場合

と比べて 13.7ナノ秒増加している．これは，中断可能関数が引数の型を調べて条件分岐を

行う必要があるためである．スレッドのコンテキスト切替えには 1回あたり 252.7ナノ秒を

要した．

ここまで述べたように，提案機構を用いたコード変換にはオーバヘッドが存在する．これ

が問題となるのかは，アプリケーションに依存する．

4.2.1 項で述べた，サーバからのファイルのダウンロードを行うようなアプリケーション

においては，このオーバヘッドは実用上の問題とならない．これは，サーバからの応答の受

信においては，通信遅延がミリ秒から秒オーダと，提案機構のオーバヘッドと比べて大きい

1 [Suspendable]

2 function run():void {

3 //ループ回数を変更し，処理の粒度を調整する
4 for (var i:int = 0; i < 1000; i++) {

5 heavyTask();

6 }

7 }

8 //重い処理を，中断可能関数ではなく通常の関数で実装する
9 function heavyTask():void {

10 //ループ回数を変更し，処理の粒度を調整する
11 for (var i:int = 0; i < 100000; i++) {

12 //重い処理
13 }

14 }

図 18 提案手法を用いた CPU 時間を消費する処理を行うプログラムのチューニング
Fig. 18 Tuning a program which consumes cpu time using the proposed method.

ためである．

4.2.2 項で述べた，アニメーションを行うアプリケーションにおいては，このオーバヘッ

ドは問題となる場合とならない場合がある．たとえば，毎秒 30回の頻度で画面を更新する

場合，およそ 33ミリ秒ごとに 1枚の画像を作る必要がある．つまり，ActionScriptの実行

時間とランタイムが描画処理を行う時間の合計が 33ミリ秒より短ければ，提案機構のオー

バヘッドは問題とならない．逆に，変換処理によってこの合計が 33ミリ秒よりも長くなっ

てしまう場合は，以下に述べる CPU時間を消費するアプリケーションと同じ問題が生じる．

4.2.3 項で述べた，CPU時間を消費するアプリケーションにおいては，このオーバヘッド

は問題となる．たとえば，図 12 に示すようにすべての処理を中断可能関数で行うプログラ

ムを記述した場合，実行時間が著しく増大してしまうことがある．

提案機構を利用して CPU時間を消費するアプリケーションを記述する開発者は，この問

題を回避するために図 18 に示すようなチューニングを行うことができる．図 18 のプログ

ラムでは，図 12 のプログラムのうち 3 カ所を修正している．うち 1 カ所は，一部の関数

の通常の関数へと変更することである（図 12 中の 7行目）．これにより，中断可能関数を

呼び出すオーバヘッドを抑えることができる．残りの 2カ所は，ループ回数の変更である

（図 18 中の 3行目と 10行目）．これにより，1回の中断処理あたりに行う処理の粒度を変

更し，高コストである中断処理が実行される回数を調整することができる．

6.3 コードサイズの増加

本手法は，ActionScriptのバイトコード中にプログラムを挿入する．このため，本手法

表 6 コード変換によるバイトコード命令数の増加
Table 6 Code size increase by bytecode translation.

プログラム コード変換前 コード変換後 比率
サーバとの通信（図 10） 38 241 6.3

アニメーション（図 11） 32 169 5.3

CPU 時間消費（図 12） 45 269 6.0

を利用すると，プログラムのサイズが増加し，SWFファイルをダウンロードするための通

信時間が増大する．4.2節で取り上げたプログラムを対象としてどのくらい増加するかを調

べた．結果を表 6 に示す．このように，通常の処理（計算や分岐）が少なく，中断可能に関

する処理が多い関数では，コード変換により，コードサイズは 5.3倍から 6.3倍になった．

我々は，このようなコードサイズの増加は実際の局面では大きな問題にならないと考えて

いる．第 1の理由は，プログラム全体の中でコードサイズが増加するのは中断可能関数のみ

であり，通常の関数は変化しないからである．たとえば，中断可能関数の割合が全体の 10%

であれば，それが 6倍になったとしても，プログラム全体サイズは 1.5倍にしかならない．

さらに，挿入されるコードも類似のものが多く，データ圧縮の効果が大きいと期待される．

第 2の理由は，Flashのアプリケーションが多くの場合，プログラム以外に画像や音声など

のデータを内部に含むか，後にサーバからダウンロードするからである．そのような場合，

大きなデータにまぎれてプログラムの部分の増加を利用者が気がつくことは少ないと思わ

れる．

7. お わ り に

本研究では，ActionScriptの処理系の上でコルーチンに基づくマルチスレッドを実現し，

開発者が ActionScript を用いてマルチスレッドプログラミングを行うことを可能とした．

マルチスレッドの実現のために，本研究では開発者が記述した関数に対してバイトコード

レベルのコード変換を行って中断可能関数を生成するためのコード変換プログラムを実装

した．中断可能関数は，継続オブジェクトに自身の実行状態を保存することによって，実行

を中断した後に再開することができる関数である．また，本研究ではランタイムライブラ

リを実装し，中断可能関数をマルチスレッド的に利用することを可能とした．提案機構を

用いることにより，開発者は複数のイベントハンドラを定義することなくサーバとの通信，

アニメーション，CPUを消費する処理を記述することができる．また，本研究では提案機

構の実装を行い，提案機構を利用することによって生じるオーバヘッドを測定した．実験環

境では，ループ 1回あたり 9.1ナノ秒，通常の関数呼び出しで 18.1ナノ秒のオーバヘッド

があった．中断可能関数の呼び出しでは，さらに 13.7ナノ秒のオーバヘッドがあった．こ

のようなオーバヘッドは，サーバと通信する処理を行うアプリケーションでは問題にならな

い．CPU時間を消費するアプリケーションでは，問題になる．この問題は，中断可能関数

において 1回の処理の粒度を調整することで解決できることを述べた．

今後の課題としては，ランタイムライブラリの改良があげられる．特に，以下の 2点につ

いて改良を行いたいと考えている．1つは，中断可能関数がループ文の中で中断処理を行う

頻度の最適化である．現在の実装では，中断処理を行う頻度は実行されるループ 100回あ

たり 1回と固定である．システム時刻からスレッドの実行時間を計測することにより，この

頻度を適切に調整する機能を追加する．もう 1つは，スレッド間の通信や同期などを行うた

めの APIの整備である．また，既存の大きなアプリケーションを再実装し，提案手法の有

用性を示す．

